


INTRODUCCIÓN

Vamos a analizar las posibilidades de MSWLogo para crear “ventanas” propias en las que podamos colocar distintos tipos de “controles”, como botones que ejecuten una determinada acción al pulsar sobre ellos, listas en las que podamos seleccionar un determinado valor, ventanas para introducir y editar texto, barras de desplazamiento, etc.

En los siguientes apartados vamos a ver como usar éstas ventanas para realizar un programa que resuelva operaciones con fracciones. En la figura adjunta vemos su aspecto. Introduciremos, desde el teclado, dos fracciones y veremos el resultado de la operación seleccionada al pinchar sobre el botón “Resultado”.




CREAR UNA VENTANA

En primer lugar, tenemos que crear la ventana principal, en la que iremos colocando los objetos y controles necesarios para nuestro programa. Esto se hace con la instrucción `creaventana`

creaventana padre nombre [título] posx posy ancho alto [instrucciones]
 Crea una ventana que se utiliza como marco en el que se pueden añadir otros objetos de ventana o controles.

- padre Es el nombre de la ventana que va a contener la que vamos a crear. Si ésta es la primera ventana creada podemos utilizar como padre una palabra en blanco (" ").
- nombre Es una palabra que identifica a la ventana creada (no es el texto que aparece en la parte superior).
- título Es el texto que aparecerá en la parte superior de la ventana.
- posx posy Son las coordenadas del vértice superior izquierdo de la ventana (el origen de coordenadas es la esquina superior izquierda de la ventana padre)
- ancho alto Son las medidas de la nueva ventana.
- instrucciones Son un conjunto de instrucciones, o un procedimiento, que se ejecutan cuando se crea la ventana.

Ejemplo: Para crear la ventana principal de nuestro programa de fracciones utilizamos la instrucción `creaventana " "miventana [Operaciones con fracciones] 5 30 250 110 []`



borraventana nombre

Cierra la ventana indicada en nombre, incluidos todos sus controles y ventanas hijas.

Ejemplo: `borraventana "miventana`

SELECCIONAR UN ELEMENTO DE UNA LISTA DE OPCIONES

En nuestro programa de fracciones, tenemos que seleccionar la operación a realizar de una lista que aparece en un control de la ventana, que recibe el nombre de "listbox". Para trabajar con estos controles tenemos que utilizar las siguientes instrucciones:

crealistbox padre nombre posx posy ancho alto

Crea un control listbox, formado por varias palabras en columna, del que podremos seleccionar una de ellas.

padre Es el nombre de la ventana que lo contiene.

nombre Nombre que le damos al control listbox.

posx posy Son las coordenadas del vértice superior izquierdo del listbox (el origen de coordenadas es la esquina superior izquierda de la ventana padre).

ancho alto Las dimensiones del control listbox.

Lógicamente tenemos que introducir en la lista el nombre de los elementos que, posteriormente, podremos seleccionar, para ello se utiliza la siguiente instrucción:

añadecadenalistbox nombre elemento

Añade una nueva palabra al listbox

nombre Nombre del control listbox en el que queremos añadir un elemento.

elemento Palabra que se añade al control listbox

Ejemplo: Vamos a crear un listbox y a introducir los nombres de las cuatro operaciones básicas (suma, diferencia, producto y división).

```
crealistbox "miventana "milista 70 25 75 50
añadecadenalistbox "milista "suma
añadecadenalistbox "milista "diferencia
añadecadenalistbox "milista "producto
añadecadenalistbox "milista "division
```

Para acceder, en cualquier momento, al elemento que hayamos seleccionado en un listbox usamos la instrucción:


leeseleccionlistbox nombre

Devuelve una lista con el elemento seleccionado del listbox. Puede ser usado directamente en el programa o guardarlo en una variable.

nombre Nombre del listbox.

Ejemplo: Lee el elemento seleccionado en el control del ejemplo anterior y lo guarda en la variable "operación. Como leeseleccionlistbox devuelve una lista, tenemos que sacar su primer elemento con la instrucción primero

```
haz "operacion pri leeseleccionlistbox "milista
```

borralistbox nombre


Cierra el control listbox cuyo nombre se indica.

habilitarlistbox nombre estado

Si estado es verdadero habilita el control listbox cuyo nombre se indica. Si estado es falso lo deshabilita.

INTRODUCIR DATOS DESDE EL TECLADO: CONTROLES COMBOBOX Y EDITA

Un control combobox son dos controles en uno (listbox y edita). Permite seleccionar un elemento de una lista dada (control listbox) o introducir, desde el teclado, la palabra, número o lista que deseamos (control edita) sin necesidad de que esté incluida en el listbox. Si deseamos crear únicamente un control edita tenemos que crear un combobox cuya altura sea suficientemente pequeña para que no quepa el listbox (dar altura 12).



creacombobox padre nombre posx posy ancho alto

Crea un control combobox, formado por un control listbox y un control edita.

padre Es el nombre de la ventana que lo contiene.


nombre Nombre que le damos al control combobox.

posx posy Son las coordenadas del vértice superior izquierdo del combobox (el origen de coordenadas es la esquina superior izquierda de la ventana padre).

ancho alto Las dimensiones del control combobox.

Ejemplo: Siguiendo con nuestro programa de fracciones vamos a crear dos controles edita que permitan introducir las dos fracciones con las que hay que hacer la operación seleccionada. Vamos a escribir el numerador y el denominador separados por un espacio en blanco (en la figura hemos escrito las fracciones 1/2 y -4/3).

```
creacombobox "miventana" "edita1" 10 25 40 12
creacombobox "miventana" "edita2" 10 55 40 12
```



Para transferir los datos introducidos en el control al programa que estamos realizando se utiliza la instrucción `leetextocombobox`:

leetextocombobox nombre

Devuelve, en forma de una lista, el contenido del control edita que está en el combobox cuyo nombre se indica.

nombre Nombre del combobox.

Ejemplo: Vamos a guardar en sendas variables el numerador y el denominador de las fracciones que hayamos escrito en los respectivos controles edita que hemos creado en el ejemplo anterior. Como `leetextocombobox` devuelve una lista tenemos que indicar el primer y el último elemento, para separar el numerador del denominador.

```
haz "n1" pri leetextocombobox "edita1"
haz "d1" ul leetextocombobox "edita1"
haz "n2" pri leetextocombobox "edita2"
haz "d2" ul leetextocombobox "edita2"
```

Otras instrucciones relacionadas con este control son:

borracombobox nombre

Borra el control combobox cuyo nombre se indica.

habilitarcombobox nombre estado

Si estado es verdadero habilita el control combobox identificado por nombre. Si estado es falso lo deshabilita.

pontextocombobox nombre texto

Escribe, dentro del control edita correspondiente al combobox identificado por "nombre", la palabra indicada en "texto".

añadelineacombobox nombre elemento


Añade, dentro del control listbox correspondiente al combobox identificado por "nombre", la lista indicada en texto.

borralineacombobox nombre indice

Borra el elemento número "indice" del listbox incluido en el combobox identificado por nombre (el índice comienza en cero).

PONER TEXTO Y ETIQUETAS EN LAS VENTANAS: CREAESTATICO.

En la ventana adjunta hemos colocados dos etiquetas (Fraccion 1 y Fraccion 2) sobre los controles edita creados en el apartado anterior. También, hemos creado una etiqueta (---) en la que el programa escribirá el resultado de las operaciones. Estas etiquetas y textos se hacen con controles estáticos:

**creaestatico** padre nombre [texto] posx posy ancho alto

Crea un control "estático", que se utiliza para mostrar un texto.

padre Es el nombre de la ventana que lo contiene.

nombre Nombre que le damos al control estático.

texto Es el texto (lista en formato de Logo) que aparecerá en el control. Posteriormente podrá ser modificado con la instrucción actualizaestatico.

posx posy Son las coordenadas del vértice superior izquierdo del control estatico (el origen de coordenadas es la esquina superior izquierda de la ventana padre).

ancho alto Las dimensiones del control estatico.

Ejemplo: Las etiquetas que aparecen en la figura anterior se han creado con las siguientes instrucciones:

```
creaestatico "miventana "Fraccion1 [Fraccion 1]13 16 35 12
creaestatico "miventana "Fraccion2 [Fraccion 2]13 46 35 12
creaestatico "miventana "Fraccion3 [-- -- --]185 43 35 12
```

borraestatico nombre


Borra el control estático cuyo nombre se indica.

actualizaestatico nombre texto

Sustituye el contenido del control estático identificado por "nombre" con un nuevo "texto".

EJECUTAR PROCEDIMIENTOS AL PINCHAR EN UN "BOTON"

En nuestro programa de operaciones con fracciones hemos añadido dos "botones": Resultado y Salir. Al pinchar en Resultado se ejecutarán los procedimientos necesarios para hacer el cálculo de la operación seleccionada. Al pinchar en Salir se cerrará el programa y la ventana correspondiente. Estos elementos se crean con los controles "botón":



creaboton padre nombre etiqueta posx posy ancho alto [instrucciones]

Creará un botón que se utiliza para provocar eventos, es decir, ejecutar una serie de instrucciones o un procedimiento cuando se pincha sobre él.

padre Es el nombre de la ventana que lo contiene.

nombre Nombre que le damos al control estático.

etiqueta Es el texto (lista en formato de Logo) que aparecerá en el botón.

posx posy Son las coordenadas del vértice superior izquierdo del control estatico (el origen de coordenadas es la esquina superior izquierda de la ventana padre).

ancho alto Las dimensiones del control estatico.

instrucciones lista de instrucciones o procedimientos a ejecutar.

Ejemplo: Para crear los botones Resultado y Salir hemos utilizado las siguientes líneas de código:

```
creaboton "miventana "Resultado "Resultado 170 25 50 12 [calcula_resultado]
creaboton "miventana "Salir "Salir 170 75 50 12 [borraventana "miventana]
```

Cuando pinchamos sobre el botón Resultado se ejecutará el procedimiento calcula_resultado, cuando lo hagamos sobre el botón Salir se ejecuta la instrucción borraventana.

borraboton nombre

Borra el botón correspondiente al nombre indicado.

habilitarboton nombre estado

Si estado es verdadero habilita el control botón cuyo nombre se indica. Si estado es falso lo deshabilita.

actualizaboton nombre texto

Sustituye la etiqueta del control boton, cuyo nombre se indica, por otra nueva expresada en texto.

LOS PROCEDIMIENTOS PRINCIPALES DE OPERACIONES CON FRACCIONES

El procedimiento inicial de nuestro programa se denomina fracciones y es el encargado de crear la ventana y los controles que hay en su interior. Es el que hemos ido desglosando en los ejemplos de los apartados anteriores. A continuación se muestra completo:

```
para fracciones
;-----
;crea la ventana principal
;-----
creaventana "main "miventana [Operaciones con fracciones] 5 30 250 110 []
;-----
;crea la lista con las operaciones que se pueden realizar
```

```

;-----
crealistbox "miventana "milista 70 25 75 50
añadecadenalistbox "milista "suma
añadecadenalistbox "milista "diferencia
añadecadenalistbox "milista "producto
añadecadenalistbox "milista "division
;-----
;crea los encabezados. Fraccion3 es en el resultado
;-----
creaestatico "miventana "Fraccion1 [Fraccion 1]13 16 35 12
creaestatico "miventana "Fraccion2 [Fraccion 2]13 46 35 12
creaestatico "miventana "Fraccion3 [-- -- --]185 43 35 12
;-----
;crea las ventanas de edicion de los dos vectores
;-----
creacombobox "miventana "combol 10 25 40 12
creacombobox "miventana "combo2 10 55 40 12
;-----
;crea los botones Resultado y Salir
;-----
creaboton "miventana "Resultado "Resultado 170 25 50 12 [calcula_resultado]
creaboton "miventana "Salir "Salir 170 75 50 12 [borraventana "miventana]
fin

```

Este procedimiento, una vez creada la ventana y los controles, finaliza y MSWLogo queda a la espera de que se pinche en alguno de los botones que hemos creado. Si pulsamos en el botón Resultado se ejecuta el procedimiento `calcula_resultado`, que es el que se encarga de obtener las fracciones introducidas en los controles edita, ver que operación hay que realizar, hacer dicha operación, simplificar el resultado y mostrarlo en el control estático correspondiente:

```

para calcula_resultado
;-----
;leemos el numerador y denominador de cada fraccion
;-----
haz "n1 pri leetextocombobox "combol
haz "d1 ul leetextocombobox "combol
haz "n2 pri leetextocombobox "combo2
haz "d2 ul leetextocombobox "combo2
;-----
;leemos la operacion a realizar y pasamos a su calculo
;-----
haz "operacion pri leeseleccionlistbox "milista
si :operacion = "suma [calcula_suma]
si :operacion = "diferencia [calcula_diferencia]
si :operacion = "producto [calcula_producto]
si :operacion = "division [calcula_division]
;-----
;mostramos el resultado
;-----
simplifica_resultado
saca_resultado
fin

```

El siguiente paso es hacer los procedimientos para resolver la operación seleccionada: `calcula_suma`, `calcula_diferencia`, `calcula_producto` y `calcula_división`. Antes vamos a analizar las instrucciones aritméticas de MSWLogo.

INSTRUCCIONES ARITMÉTICAS

Logo dispone de un conjunto amplio de instrucciones para realizar operaciones aritméticas (se pueden consultar en la ayuda del programa). Vamos a ver, a continuación, algunas de ellas:

numero1 + numero2

Devuelve la suma de los números indicados

numero1 - numero2

Devuelve la diferencia del valor numérico de las entradas

numero1 * numero2

Devuelve el producto del valor numérico de las entradas

numero1 / numero2

Devuelve el cociente de la división de los valores numéricos de las entradas

Hay que tener en cuenta que con el resultado de una operación hay que hacer alguna actuación (por ejemplo, escribe), si no Logo nos dará un mensaje de error.

Ejemplo: 5 + 7 La respuesta es: no me has dicho que hacer con 12
 es 5 + 7 12

Podemos incluir varias operaciones seguidas, teniendo en cuenta que primero se ejecutan las multiplicaciones y divisiones y posteriormente las sumas y las restas. También se pueden usar los paréntesis para darle prioridad a una determinada operación:

Ejemplos:

es 8/2+3	7	primero hace la división y después la suma
es 8/(2 + 3)	1.6	primero hace la suma y después la división
es 2+4*5+3/2	23.5	primero multiplicación y división y después sumas
es (2+4*5+3)/2	125	multiplicación, sumas y finalmente división

resto numero1 numero2

Devuelve el resto de dividir numero1 entre numero2; ambos números tienen que ser enteros y el resultado es otro entero con el mismo signo que tenga numero1.

potencia numero1 numero2

Devuelve numero1 elevado a la potencia numero2

raizcuadrada numero1**rc numero1**

Devuelve la raíz cuadrada de numero1 (como sabéis este número no puede ser negativo).

Ejemplo: potencia 2 3 8
 rc 25 5

A continuación se muestran los procedimientos `calcula_suma`, `calcula_diferencia`, `calcula_producto` y `calcula_división`. El numerador y el denominador del resultado de cada operación se ha almacenado en las variables "nr" y "dr"

```
para calcula_suma
haz "nr (:n1 * :d2) + (:n2 * :d1)
haz "dr :d1 * :d2
fin
```

```
para calcula_diferencia
haz "nr (:n1 * :d2) - (:n2 * :d1)
haz "dr :d1 * :d2
fin
```

```
para calcula_division
haz "nr :n1 * :d2
```

```
haz "dr :d1 * :n2
fin
```

```
para calcula_producto
haz "nr :n1 * :n2
haz "dr :d1 * :d2
fin
```

SIMPLIFICAR Y MOSTRAR EL RESULTADO

Para simplificar el resultado dividimos el numerador y el denominador por su máximo común divisor.

```
para simplifica_resultado
haz "m mcd :nr :dr
haz "nr :nr / :m
haz "dr :dr / :m
fin
```

La obtención del mcd se realiza mediante el procedimiento recursivo adjunto, que asumimos como una instrucción más de MSWLogo, sin entrar en su análisis.

```
para mcd :a :b
; máximo común divisor
sisino :b=0 [dev :a] [dev mcd :b resto :a :b]
fin
```

El resultado se muestra actualizando el estático "Fraccion3. Hemos tenido en cuenta que si el numerador y el denominador son negativos el resultado es positivo y, por tanto, hemos quitado los signos menos. En el caso de que el numerador sea positivo y el denominador negativo eliminamos el signo menos de éste y lo ponemos en el numerador, para que el signo aparezca al inicio de la fracción.

```
para saca_resultado
si y :nr<0 :dr<0 [haz "nr mp :nr haz "dr mp :dr]
si y :nr>0 :dr<0 [haz "nr palabra "-" :nr haz "dr mp :dr]
actualizaestatico "Fraccion3 (lista :nr "/" :dr)
fin
```

